Mx script to fit a moderated second-order common factor model

```
#define nvar 12          ! number of observed variables excluding the moderator
#define nfac 3           ! number of factors
#define ndef 1           ! number of 'definition variables' (i.e., number of moderators)

#ngroups 1               ! number of groups

g1: model group 1
da ni =13                ! number of observed variables including the moderator
mi=-99

rec file = data.dat      ! specify the observed data file

labels                   !specify labels for the 13 observed variables
        calendar
        cube
        vocabulary
        surface
        form
        vocabulary
        paper
        object
        identical
        newcastle
        spelling
        mazes
        m1               ! note that we label the moderator 'm1'

definition               ! specify variable 'm1' as the definition (or moderator) variable
m1;

matrices;
A full 1 nvar fi         ! intercept baseline parameters
B full 1 nvar fi         ! intercept moderation parameters

C full 1 nvar fi         ! subtest residual variance baseline parameters
D full 1 nvar fi         ! subtest residual variance moderation parameters

E full nvar nfac fi      ! factor loadings baseline parameters
F full nvar nfac fi      ! factor loadings moderation parameters

G full 1 nfac fi         ! first-order residual variance baseline parameters
H full 1 nfac fi         ! first-order residual variance moderation parameters

I full nfac 1 fi         ! second order factor loadings baseline parameters
J full nfac 1 fi         ! second order factor loadings moderation parameters

K full 1 1 fi            ! second order factor variance baseline parameter
L ful 1 1 fi             ! second order factor variance moderation parameter
```

```
M fu 1 1                    ! M will contain the definition (or moderation) variable

end matrices;


cov (E+F@M) *                                         ! See explanation below
(  (I+J@M) * \exp(K+L@M) *
(I+J@M)' + \v2d((G+H@M))  ) *
(E+F@M)'   + \v2d(\exp(C+D@M)) /
```

! This formula gives the predicted covariance matrix conditional on the moderator. In
! standard (i.e., unmoderated) factor analysis, this formula is given by
!
! $$\Sigma = \Lambda\ (\Gamma\varphi\Gamma^T+\Psi)\Lambda^T + \Theta \quad (1)$$
!
! In moderated factor analysis, we model each of these parameters as a function of the
! moderator, M. For the loadings (i.e., first-order loadings, Λ, and second-order
! loadings, Γ) we use linear functions, e.g., for the first-order loadings
!
! $$\Lambda_M = \Lambda_0+\Lambda_1\times M,$$
!
! where subscript M denotes ‘conditional on M’, subscript 0 denotes ‘baseline
! parameter’, and subscript 1 denotes ‘moderation parameter’.
! For the variances (i.e., subtest residual variances, Θ, first-order residual variances, Ψ,
! and second-order factor variance, φ) we use exponential functions, e.g., for the
! subtest residual variances:
!
! $$\Theta_M = \exp(\ \Theta_0+\Theta_1\times M\ )$$
!
! see Hessen & Dolan (2009) and Bauer & Hussong (2009).
!
! Now we introduce the moderation effects in all parameter of (1) and we obtain:
!
! $$\Sigma_M = \Lambda_M\ (\Gamma_M\varphi_M\ \Gamma_M{}^T+\Psi_M)\Lambda_M{}^T+\Theta_M$$
!
! $$= (\Lambda_0+\Lambda_1\times M)$$
! $$[\ (\Gamma_0+\Gamma_1\times M)\ \ \exp(\varphi_0+\varphi_1\times M)\ (\Gamma_0+\Gamma_1\times M)^T + \exp(\Psi_0+\Psi_1\ \times M)\ ]$$
! $$(\Lambda_0+\Lambda_1\times M)^T +$$
! $$\exp(\ \Theta_0+\Theta_1\times M\ )$$
!
! To implement this in Mx, we use matrices A to M (as specified in the syntax above)
! for the matrices Λ_0, Λ_1, Ψ_0, Ψ_1, etc. Thus, $\Lambda_0 + \Lambda_1\times M$ translates to E+F@M (we use
!a kronecker product, @, as M is a 1x1 matrix), $\exp(\Psi_0+\Psi_1\ \times M)$ translates to
! exp(G+H@M), etc.

```
me A+B@M /
```

! As we model a single group only, the model of the means only consists of the
! intercept parameters. Here we specify the intercepts (υ) that are also moderated
! according to a linear function, i.e.,
!
! $\upsilon_M = \upsilon_0 + \upsilon_1 \times M$
!
! As matrix A corresponds to υ_0 and B to υ_1 the formula gets A + B@M

```
sp M                    ! The definition variable (or moderator) ‘m1’ is stored in matrix
m1                      ! M.

!###########################
!#      Intercepts       ###
!##########################

sp A                                ! estimate all baseline parameters
1 2 3 4 5 6 7 8 9 10 11 12

sp B
21 22 23 24 25 26 27 28 29 30 31 32    ! estimate all moderation parameters

!###########################
!#      Res Var          ###
!##########################

sp C                                ! estimate all baseline parameters
41 42 43 44 45 46 47 48 49 50 51 52

sp D                                ! estimate all moderation parameters
61 62 63 64 65 66 67 68 69 70 71 72

!###########################
!#      Loadings         ###
!##########################

sp E                          ! estimate some of the baseline parameters
0     0     0
81    0     0
82    0     0
83    84    0
0     0     0
0     86    0
0     87    0
0     88    0
0     89    0
0     85    90
```

```
0         0         91
0         0         0

ma E                                  ! fix some baseline parameters to 1, and provide
1         0         0                 ! starting values for those that are estimated (see
1         0         0                 ! above)
1         0         0
1         1         0
0         1         0
0         1         0
0         1         0
0         1         0
0         1         0
0         1         1
0         0         1
0         0         1



!###########################
!#        Psi                ###
!###########################

sp G                                  ! estimate all baseline parameters
101 0 103

sp H                                  ! estimate all moderation  parameters
111 112 113

!###########################
!#        2nd order loadings      ###
!###########################

sp I                                  ! estimate all baseline parameters
121 126 122

sp J                                  ! do not estimate the moderation parameters
0 0 0

ma I                                  ! starting values for the baseline parameters
1 1 1

!###########################
!#    2nd order factor  var       ###
!###########################

sp K                                  ! estimate the baseline parameter
0
```

```
sp L                                        ! estimate the moderation parameter
141

!##################################################

op
end
```